In[1]:=
```
(* Malgorzata Mikosz, Andrzej Weber
 Triality in so(4,4). *)
```

In[2]:=
```
(* Two generic elements of so(4,4) *)
DD = Table[x[i + 10 j], {j, 1, 8}, {i, 1, 8}];
Do[DD[[9 - i, i]] = 0, {i, 1, 8}];
Do[Do[DD[[i, j]] = -DD[[9 - j, 9 - i]], {i, 10 - j, 8}], {j, 2, 8}]
CC = Table[y[i + 10 j], {j, 1, 8}, {i, 1, 8}];
Do[CC[[9 - i, i]] = 0, {i, 1, 8}];
Do[Do[CC[[i, j]] = -CC[[9 - j, 9 - i]], {i, 10 - j, 8}], {j, 2, 8}]
```

In[7]:=
```
(* The triality automorphism *)
triality[A_] := {{1 / 2 (A[[1, 1]] + A[[2, 2]] + A[[3, 3]] + A[[4, 4]]),
    A[[3, 5]], -A[[2, 5]], A[[1, 5]], A[[2, 6]], A[[1, 6]], A[[1, 7]], 0},
   {A[[5, 3]], 1 / 2 (A[[1, 1]] + A[[2, 2]] - A[[3, 3]] - A[[4, 4]]),
    A[[2, 3]], -A[[1, 3]], A[[2, 4]], A[[1, 4]], 0, -A[[1, 7]]},
   {-A[[5, 2]], A[[3, 2]], 1 / 2 (A[[1, 1]] - A[[2, 2]] + A[[3, 3]] - A[[4, 4]]),
    A[[1, 2]], A[[3, 4]], 0, -A[[1, 4]], -A[[1, 6]]},
   {A[[5, 1]], -A[[3, 1]], A[[2, 1]], 1 / 2 (-A[[1, 1]] + A[[2, 2]] + A[[3, 3]] - A[[4, 4]]),
    0, -A[[3, 4]], -A[[2, 4]], -A[[2, 6]]}, {A[[6, 2]], A[[4, 2]], A[[4, 3]], 0,
    1 / 2 (A[[1, 1]] - A[[2, 2]] - A[[3, 3]] + A[[4, 4]]), -A[[1, 2]], A[[1, 3]], -A[[1, 5]]},
   {A[[6, 1]], A[[4, 1]], 0, -A[[4, 3]], -A[[2, 1]],
    1 / 2 (-A[[1, 1]] + A[[2, 2]] - A[[3, 3]] + A[[4, 4]]), -A[[2, 3]], A[[2, 5]]},
   {A[[7, 1]], 0, -A[[4, 1]], -A[[4, 2]], A[[3, 1]], -A[[3, 2]],
    1 / 2 (-A[[1, 1]] - A[[2, 2]] + A[[3, 3]] + A[[4, 4]]), -A[[3, 5]]},
   {0, -A[[7, 1]], -A[[6, 1]], -A[[6, 2]], -A[[5, 1]], A[[5, 2]],
    -A[[5, 3]], 1 / 2 (-A[[1, 1]] - A[[2, 2]] - A[[3, 3]] - A[[4, 4]])}};
MatrixForm[triality[
  DD]]
```

Out[8]//MatrixForm=

$$\begin{pmatrix}
\frac{1}{2}(x[11]+x[22]+x[33]+x[44]) & x[35] & -x[25] & x[15] & x[26] & x[16] & x[17] & \\
x[53] & \frac{1}{2}(x[11]+x[22]-x[33]-x[44]) & x[23] & -x[13] & x[24] & x[14] & 0 & -\\
-x[52] & x[32] & \frac{1}{2}(x[11]-x[22]+x[33]-x[44]) & x[12] & x[34] & 0 & -x[14] & -\\
x[51] & -x[31] & x[21] & \frac{1}{2}(-x[11]+x[22]+x[33]-x[44]) & 0 & -x[34] & -x[24] & -\\
x[62] & x[42] & x[43] & 0 & \frac{1}{2}(x[11]-x[22]-x[33]+x[44]) & -x[12] & x[13] & -\\
x[61] & x[41] & 0 & -x[43] & -x[21] & \frac{1}{2}(-x[11]+x[22]-x[33]+x[44]) & -x[23] & \\
x[71] & 0 & -x[41] & -x[42] & x[31] & -x[32] & \frac{1}{2}(-x[11]-x[22]+x[33]+x[44]) & -\\
0 & -x[71] & -x[61] & -x[62] & -x[51] & x[52] & -x[53] & \frac{1}{2}(-x[11]-x[
\end{pmatrix}$$

In[9]:=
```
(* The order of the triality is equal to 3 *)
Simplify[triality[triality[triality[DD]]]] == DD
```

Out[9]= True

In[10]:=
```
(* Definition of the Lie bracket *)
bracket[A_, B_] := Simplify[A.B - B.A]
```

In[11]:=
```
(* Triality preserves bracket *)
Simplify[bracket[triality[CC], triality[DD]]] == Simplify[triality[bracket[CC, DD]]]
```

Out[11]= True

In[12]:=
```
(* G2 - The form of matrices fixed by the triality *)
GG = DD /. (SolveAlways[triality[DD] == DD, {}][[1]]);
MatrixForm[GG]
```

Out[13]//MatrixForm=

$$\begin{pmatrix} x[22]+x[33] & x[35] & -x[25] & x[26] & x[26] & x[16] & x[17] & 0 \\ x[53] & x[22] & x[23] & x[25] & x[25] & x[26] & 0 & -x[17] \\ -x[52] & x[32] & x[33] & x[35] & x[35] & 0 & -x[26] & -x[16] \\ x[62] & x[52] & x[53] & 0 & 0 & -x[35] & -x[25] & -x[26] \\ x[62] & x[52] & x[53] & 0 & 0 & -x[35] & -x[25] & -x[26] \\ x[61] & x[62] & 0 & -x[53] & -x[53] & -x[33] & -x[23] & x[25] \\ x[71] & 0 & -x[62] & -x[52] & -x[52] & -x[32] & -x[22] & -x[35] \\ 0 & -x[71] & -x[61] & -x[62] & -x[62] & x[52] & -x[53] & -x[22]-x[33] \end{pmatrix}$$

In[14]:=
```
HH = GG /. {x[61] → a11, x[71] → a12, x[22] → t2, x[32] → a9,
   x[52] → a8, x[62] → a10, x[23] → a7, x[33] → t3, x[53] → a6, x[25] -> a2,
   x[35] → a1, x[16] → a4, x[26] → a3, x[17] → a5}; MatrixForm[HH]
```

Out[14]//MatrixForm=

$$\begin{pmatrix} t2+t3 & a1 & -a2 & a3 & a3 & a4 & a5 & 0 \\ a6 & t2 & a7 & a2 & a2 & a3 & 0 & -a5 \\ -a8 & a9 & t3 & a1 & a1 & 0 & -a3 & -a4 \\ a10 & a8 & a6 & 0 & 0 & -a1 & -a2 & -a3 \\ a10 & a8 & a6 & 0 & 0 & -a1 & -a2 & -a3 \\ a11 & a10 & 0 & -a6 & -a6 & -t3 & -a7 & a2 \\ a12 & 0 & -a10 & -a8 & -a8 & -a9 & -t2 & -a1 \\ 0 & -a12 & -a11 & -a10 & -a10 & a8 & -a6 & -t2-t3 \end{pmatrix}$$

In[15]:=
```
v = {b1, b2, b3, b4, b5, b6, b7, b8};
```

In[16]:=
```
(* The vector fixed by G2 *)
v /. SolveAlways[{v}.GG == 0, Variables[GG]]
```

Out[16]=
```
{{0, 0, 0, -b5, b5, 0, 0, 0}}
```

In[17]:=
```
(* The basis of (e5-e4)^\perp *)
f[i_] := (vector = IdentityMatrix[8][[i]]; If[i == 4, vector[[5]] = 1]; vector)
(* The vector f[4] is equal to e4+e5 *)
```

In[18]:=
```
(* The basis of 3 forms *)
base = Complement[Range[1, 8], {5}];
base3 = Subsets[base, {3}]
```

Out[19]=
```
{{1, 2, 3}, {1, 2, 4}, {1, 2, 6}, {1, 2, 7}, {1, 2, 8}, {1, 3, 4}, {1, 3, 6},
 {1, 3, 7}, {1, 3, 8}, {1, 4, 6}, {1, 4, 7}, {1, 4, 8}, {1, 6, 7}, {1, 6, 8},
 {1, 7, 8}, {2, 3, 4}, {2, 3, 6}, {2, 3, 7}, {2, 3, 8}, {2, 4, 6}, {2, 4, 7},
 {2, 4, 8}, {2, 6, 7}, {2, 6, 8}, {2, 7, 8}, {3, 4, 6}, {3, 4, 7}, {3, 4, 8},
 {3, 6, 7}, {3, 6, 8}, {3, 7, 8}, {4, 6, 7}, {4, 6, 8}, {4, 7, 8}, {6, 7, 8}}
```

In[20]:=
```
(* Coefficients in the space of 3-forms *)
wyb[a_, b_] := {b[[a[[1]]]], b[[a[[2]]]], b[[a[[3]]]]}
minor[a_, v1_, v2_, v3_] := Det[wyb[a, Transpose[{v1, v2, v3}]]]
```

In[22]:=
```
(* The action of the general matrix *)
h[v1_] := HH.v1 ;
(* The action on \Lambda^3 *)
h3[v1_, v2_, v3_] := Factor[Table[
   minor[a, h[v1], v2, v3] + minor[a, v1, h[v2], v3] + minor[a, v1, v2, h[v3]], {a, base3}]]
```

In[24]:=
```
(* Fixed 3-form *)
variables = Table[c[a], {a, base3}];
virt = Sum[c[a] e[a], {a, base3}];
Factor[virt /. (SolveAlways[
      Sum[c[a] h3[f[a[[1]]], f[a[[2]]], f[a[[3]]]], {a, base3}] == 0, Variables[HH]][[1]])]
```

Out[26]=
```
-c[{3, 4, 6}] (e[{1, 4, 8}] - 2 e[{1, 6, 7}] - 2 e[{2, 3, 8}] - e[{2, 4, 7}] - e[{3, 4, 6}])
```

In[27]:=
```
Factor[% / c[{3, 4, 6}]]
```

Out[27]=
```
-e[{1, 4, 8}] + 2 e[{1, 6, 7}] + 2 e[{2, 3, 8}] + e[{2, 4, 7}] + e[{3, 4, 6}]
```